
1. Summary 

We are submitting only a small number of model runs in response to phase 2 of this study as 
it is not yet clear to us how a large number of runs would be compared given some 
ambiguities in the protocol.  

Results are submitted for two cases which differ only in the time period over which emissions 
are attributed to regions. In both cases we have used: 

• the EDGAR-HYDE v1.4 database for past greenhouse gas emissions; 
• the SRES A2 scenario for future greenhouse gas and aerosol emissions; 
• standard parameters for the carbon cycle with minimal adjustment to improve the 

simulation of present day CO2 concentrations for the emissions used; 
• fixed lifetimes for CH4 and N2O, with the lifetime value adjusted to match 

observations; 
• total aerosol radiative forcing is related to emissions using parameters determined 

from a least squares fit of the EDGAR-HYDE v1.3 database for past total sulfur 
emissions and the UKMO estimates of total aerosol forcing; 

• this approach allows us to attribute past sulfate aerosol forcing by region based on 
EDGAR-HYDE v1.3; 

• temperature change is related to radiative forcing using the impulse response function 
provided by UKMO; 

• mean sea level results are not reported. 

In Case I we have attributed emissions to regions for the period 1890 to 2100 and in Case II 
emissions are attributed for the period 1890 to 2000. All emissions outside the period of 
attribution are classed as “Unattributed”.  

The models used attribute atmospheric and climatic changes according to the source region of 
the emissions as required in the study protocol and described in more detail below. In 
addition, the changes due to total emissions are calculated separately as a check that the sum 
of regional contributions for all parameters equals the total change. 

Note that as the SRES scenarios do not include bunker fuels as a separate category, and given 
the number of other ambiguities here, we have NOT treated bunker fuel emissions as 
unattributed because that would seem likely to confuse interpretation of the difference 
between cases I and II as run here. Instead we have attributed past bunker fuel emissions from 
the EDAGR-HYDE v1.4 database on a pro-rata basis to each region. The 3 MtSyr-1 difference 
between the world total SOx emissions in the SRES and the sum of the four regional 
groupings is attributed to bunker fuel emissions (see SRES text) and so a similar approach is 
taken there to scale up regional emissions to ensure that they do match the world total. 



 

 

Results are provided for components of change classified as Unattributed, OECD90, REF, 
ASIA, ALM and the Total change, for each of: 

• cumulative emissions of CO2, CH4, and N2O 
• concentrations of CO2, CH4, and N2O 
• radiative forcing due to the three greenhouse gases and aerosols 
• temperature. 

Based on our access to the study web site, we have found MS Excel spreadsheets the most 
convenient way of viewing data provided by other participants, so have made our results 
available in that format. 

2. Determination of Emissions 

Carbon Dioxide 

As suggested in the protocol for phase 2 we have taken historical CO2 emissions from the 
EDGAR-HYDE v1.4 database and future emissions from the SRES A2 ASF marker scenario.  

Emissions are classified for each of the four regions: OECD90, REF, ASIA, and ALM in the 
terminology of SRES. These regions are explicitly identified for SRES and for the EDGAR-
HYDE database we have assumed the correspondence: 
  OECD90 =  Canada + USA + OECD Europe + Oceania + Japan 
  REF =  Eastern Europe + CIS 
  ASIA = India Region + China Region + Southeast Asia 
  ALM = Latin America + Africa + Middle East 

However, it should be noted that there is some mismatch in CO2 regional emissions at the 
overlap between the historical EDGAR-HYDE database and the SRES scenario. While the 
SRES total in 1990 agrees to better than 1% with the corresponding EDGAR-HYDE value, 
the scenario starts 13% low for the REF region and 35% high for the ALM region. We have 
used SRES values from 2000 onward and linearly interpolated between years specified in the 
databases. 

Bunker fuels are specified explicitly in the EDGAR-HYDE database but not in the SRES 
scenario. There does not appear to be any value in carrying out an attribution for bunker fuels 
separately in this case, correspondingly we have redistributed bunker fuel CO2 emissions 
across regions in the EDGAR-HYDE database by scaling regional emissions up each year to 
match the total including bunker fuels. This scaling is typically 3%. 

We also note that the Marland et al fossil fuel CO2 emissions used in Phase 1 of this study are 
generally 90 to 95% of those given in the EDGAR-HYDE database, whereas the Houghton et 
al land-use change emissions are more than twice those in EDGAR-HYDE. These differences 
provide an option for assessing sensitivity in the attributions to uncertainties in historical 
emission values, although that has not been done here. They also show that one needs to be 
careful to recognize where and when models are using different emission inputs.  

Methane 

The approach taken for CH4 emissions is parallel to that used for CO2 emissions as just 
described. 



 

 

Note that at 1990, where EDGAR-HYDE and SRES overlap, the total emissions in the 
scenario are 2.6% higher than in the historical database. Also, as for CO2, there are larger 
mismatches at the regional level, with the scenario starting 17% low for the REF region and 
16% high for the ALM region.  

We also note that the Stern and Kauffman methane emissions used in Phase 1 of the study are 
typically 20% higher than the world total in the EDGAR-HYDE database. 

Nitrous Oxide 

The approach taken for N2O emissions is also parallel to that used for CO2 and CH4 
emissions. 

The mismatch between SRES and the EDGAR-HYDE database is relatively more serious for 
N2O. The world total emissions in the scenario are 6.6% lower than in the historical database, 
and at the regional level each region differs by more than 10% from history to scenario. The 
worst case is for ALM which starts 46% low in the scenario.  

These discrepancies suggest the need for some re-examination of the emission values before 
their use in attribution, however, for the purposes of this study we have ignored the 
mismatches, used SRES values for 2000 and onward, and linearly interpolated for years that 
are not explicitly specified. 

Sulfur emissions and aerosol effects 

In order to determine an attribution of sulfate aerosol radiative forcing by emitting region we 
assume (following the TAR) that radiative forcing is proportional to emission rate. Emissions 
have been taken from the EDGAR-HYDE v1.3 database which specifies SO2 emissions for 
regions over the period 1890 to 1990. For the post 1990 period we have used the SRES A2 
ASF values for SOx emissions. These agree to within 7% in the overlap year 1990. 

To convert emissions to radiative forcing in a way that is consistent with the UKMO 
HadCM3 study, we have calculated the best fit scaling factors which when applied to world 
total emissions reproduce the direct and indirect radiative forcing effects specified by the 
UKMO. These straight line, no intercept, fits of emissions to radiative forcing are quite good 
for the direct effect but underestimate the magnitude of the indirect effect at low emissions 
and overestimate the effect at high emissions.  

We have then used these scaling coefficients to estimate the direct and indirect forcing due to 
each region’s SO2 emissions both before and after 1990. Linear interpolation in time is used 
to calculate emissions between the times specified in the databases. 

Emissions Summary 
Figure 1 shows the attribution of greenhouse gas emissions for Case 1 and figure 2 does 
likewise for Case 2. Aerosol emissions scaled to radiative forcing values are shown in the 
middle panels of Figures 6 and 7. 
 



 

 

 
Figure 1. Case I emissions for CO2, CH4 and N2O over the period 1750 to 2100 and their 
attribution. 

 

Figure 2. Case II emissions for CO2, CH4 and N2O over the period 1750 to 2100 and their 
attribution. 

3. From Emissions to Concentrations 

Carbon dioxide 

The approach taken here is to use a simple carbon cycle model with the emissions as 
described above and to make minimal adjustments to the model parameters so as to reproduce 
observed CO2 concentrations.  

The model used is the pulse response model of Joos et al (1996) with separate response 
functions applied to describe the ocean and biosphere uptake processes. Note that this is not 
the same as using a single response function for atmospheric concentrations and has been 
shown to be much more accurate in dealing with large perturbations of the carbon cycle. The 
response functions used are those developed to simulate the HILDA ocean model and a 4-box 
biosphere as described in the original publication. In addition we use a simple logarithmic 
dependence of NPP on atmospheric CO2 concentration with β = 0.287. 

To minimize non-linearities, other feedbacks have been neglected at this stage. In particular 
we do not treat temperature feedbacks on sea water CO2 solubility. 

The only parameter of the model that is adjusted from standard defaults is the effective air-sea 
exchange coefficient. The value used is 1/(14.3 yr) rather than the standard 1/(9.6 yr) and has 
been chosen to reproduce CO2 concentrations in the 1990s for the prescribed emissions 
history and to give an ocean uptake of 2.5 GtCyr-1. The biospheric uptake in the 1990s is 1.4 
GtCyr-1.  



 

 

In common with most studies using simple carbon cycle models and “bottom-up” estimates of 
CO2 emissions, the simulated concentrations are lower than observed in the late 19th and 
early 20th century. The maximum error is about 9 ppm at around 1920. However, the more 
recent part of the record, where emissions are probably better known, is reproduced quite 
well. 

A variant of the Joos pulse response model is used here in order to track the CO2 separately 
for each source region. This is straightforward except where the model relationships are non-
linear. Thus we track separate atmospheric concentrations, dissolved inorganic carbon 
concentrations in the surface ocean, air-sea fluxes, etc for each “color” of CO2 according to its 
source region.  

Nonlinearities occur in the dependence of surface ocean pCO2 on dissolved inorganic carbon 
concentration, and in the dependence of NPP on atmospheric CO2 concentration. In both cases 
we adopt the marginal attribution approach similar to that of Enting et al (1998) and attribute 
the change each year in pCO2 or in NPP according to the attributed change in the driving 
factors. The drivers for pCO2 are taken to be the previous year’s change in dissolved 
inorganic carbon and for NPP are taken to be the change in atmospheric concentrations. For 
example, the increase in pCO2 attributed to the OECD90 emissions each year is a proportion 
of the change in total pCO2 equal to the ratio of the change in OECD90 dissolved inorganic 
carbon to the change in total dissolved inorganic carbon. 

The attribution of CO2 concentrations to emissions by region as described above does not give 
identical results to the situation where those emissions are the only ones considered. However 
the differences do not appear to be large. For example Figure 3 compares the unattributed 
CO2 concentrations arising from emissions prior to 1890 in Case I with the CO2 
concentrations that would have arisen if there had been no emissions after 1890. 

Figure 3.  Comparison of CO2 concentrations due to emissions before 1890 with and without 
subsequent emissions taken into account. 

Methane 

As noted elsewhere it is not possible to match CH4 emission histories to observed changes in 
concentrations with a fixed lifetime for this species and the reasons for change in the lifetime 
are well understood. In addition the assumption of constant non-anthropogenic sources is also 
suspect. However, in order to conform to the study protocol we have used a fixed lifetime and 
fixed non-anthropogenic sources throughout. 

As in the case of CO2 above, the approach taken is to adjust the parameters of the methane 
concentration model to give a reasonably good fit to observations – despite the caveat just 
made. A fixed lifetime of 10.0 yr is obtained this way which is probably too short for the 
average perturbation lifetime over this period, however, recent values are reasonably well 
simulated.  

Nitrous oxide 

N2O has similar problems of consistency between observed concentrations, specified sources, 
and the constraint of constant lifetime and non-anthropogenic source.  

We have followed the advice attached to the study protocol and used a (perturbation) lifetime 
of 114 years as in the TAR.  

Greenhouse gas summary 

Figures 4 and 5 show the change in greenhouse gas concentrations for cases 1 and 2 
respectively.  

 

Figure 4. Change in greenhouse gas concentration for Case I with attribution to regions. 
Observed concentration for CO2 and CH4 are shown with X marks. 
 



 

 

 

Figure 5. Change in greenhouse gas concentration for Case II with attribution to regions. 

4. Calculation of Radiative Forcing 

The radiative forcing is calculated as the sum of components for CO2, CH4, N2O and sulfate 
aerosol. Each component of the forcing is attributed to an emission region as described below 
and so the total forcing is also. 

The total radiative forcing reaches 6.50 W · m-2 in 2100 for the A2 scenario which is 
comprised of 7.45 W · m-2 for the three greenhouse gases and -0.95 W · m-2 for aerosol 
forcing.  

Carbon dioxide 

The standard logarithmic formula for CO2 radiative forcing as given in the TAR is used.  

Because this is a non-linear relationship the attribution of forcing to each source region 
follows the marginal change approach used by Enting (1998) as in the variant carbon cycle 
model described above. Thus the increase in forcing attributed to OECD90 each year is the 
proportion of the total increase in forcing given by the ratio of the change in atmospheric CO2 
concentration attributed to the OECD90 region to the total change in atmospheric CO2. 

Methane and Nitrous Oxide 

The standard square root formulae for radiative forcing of CH4 and N2O as given in the TAR 
are used. 

Non-linearities are treated as for CO2 radiative forcing. 

Sulfate aerosol 

As noted in discussing sulfur emissions above we have used the UKMO estimates of direct 
and indirect radiative forcing to determine factors relating emissions in the EDGAR-HYDE 
database to forcing and then applied these to the estimates of regional emissions in the past 
and the future. 

The future scenario for aerosol forcing shown in the UKMO results does not correspond with 
the SRES A2 scenario in some respects (UKMO may have used a preliminary rather than 
final version of this scenario?). In particular the maximum in sulfur emissions in the A2 



 

 

scenario occurs in 2030 whereas the UKMO aerosol forcing reaches its maximum magnitude 
in 2050. We are using the SRES data strictly rather than trying to work backwards from the 
UKMO values to emissions. Also, perhaps due to the way we have mapped emissions to 
forcing here, the magnitude of forcing in 2100 is about 20% less than that given in the UKMO 
values. However, there is generally a reasonably good correspondence between our aerosol 
forcing and that used for the HadCM3 model run. 

The sum of direct and indirect aerosol forcing is added to that for greenhouse gases for each 
region and as the relationship between emission and forcing is linear no problems arise in 
attribution here. 

Summary of Radiative Forcing 

Figures 6 and 7 show attributed radiative forcing (including aerosol effects) for Cases I and II 
respectively. 

 

Figure 6. Radiative forcing due to greenhouse gases and aerosols and total forcing for Case I, 
showing attribution to regions. 

 

 

Figure 7. Radiative forcing due to greenhouse gases and aerosols and total forcing for Case 
II, showing attribution to regions. 

 



 

 

5. Calculation of temperature change 

We have used both a simple model simulating oceanic heat uptake and the impulse response 
function supplied by UKMO in the web pages for the study. There are differences in the rate 
of response of global mean temperature to radiative forcing with the impulse response 
function generally having the slower response.  

Here we limit results to those for the impulse response model. The total radiative forcing used 
here is incomplete, e.g. does not include forcing due to ozone and CFCs. However, to be 
consistent with our approach to models for greenhouse gas concentration we have adjusted the 
climate sensitivity parameter in the impulse response model to give a reasonable simulation of 
recent temperatures. This requires use of a climate sensitivity parameter of 4.1°C. Use of  a 
different climate sensitivity will not change the relative share of total warming attributed to 
regions.  

Figures 8 and 9 show the change in temperature attributed to regions for cases I and II 
respectively. 

 

Figure 8. Temperature change attributed to emissions by region for Case I. The left panel 
shows regional contributions as a stacked area plot. Observed temperature anomalies from 
UEA are shown as crosses after adjustment so that observed and model values have the same 
mean over the 1961 to 1990 period. The right hand panel shows individual contributions by 
region more clearly as a line plot. 



 

 

 

Figure 9. Temperature change attributed to emissions by region for Case II. The left panel 
shows regional contributions as a stacked area plot. Observed temperature anomalies from 
UEA are shown after adjustment so that observed and model values have the same mean over 
the 1961 to 1990 period. The right hand panel shows individual contributions by region more 
clearly as a line plot. 

Relative Attribution 

Because we have included sulfate aerosol radiative forcing attributed by region the total 
warming contribution from regional emissions is negative in several cases. However, by 
coincidence the contributions are positive for the attribution dates 2000, 2050 and 2100 
requested in the study. (Note we have not considered time scales to 2500 as suggested in the 
study protocol.) 

The absolute and relative warming attributed to regions is shown in Table 1 and in Figures 10 
and 11.  

Table 1. Warming in °C  attributed to regions for the two cases discussed in the text. 
 Warming in degrees C % of total 
Year Unattr OECD90 REF ASIA ALM Total Unattr OECD90 REF ASIA ALM 

Case I 
2000 0.12 0.30 0.04 0.06 0.14 0.65 18.0 45.6 6.5 8.9 21.1 
2050 0.11 0.89 0.21 0.27 0.34 1.82 6.2 48.9 11.6 14.8 18.5 
2100 0.11 1.37 0.49 1.42 1.09 4.49 2.5 30.6 10.9 31.7 24.2 

Case II 
2000 0.12 0.30 0.04 0.06 0.14 0.65 18.0 45.6 6.5 8.9 21.1 
2050 0.86 0.43 0.14 0.20 0.20 1.82 46.9 23.8 7.8 10.8 10.7 
2100 3.59 0.40 0.13 0.18 0.18 4.49 80.1 9.0 2.9 4.0 4.0 

 



 

 

 

Figure 10. Absolute and relative attribution of warming to regions for Case I. 

 

Figure 11. Absolute and relative 
attribution of warming to regions for Case II. 

Appendix – IDL Code 

The following listings show the code used with the IDL package to produce the results given 
above. 

Files “case1.pro” and “case2.pro” are script files that are executed to compile and run the 
procedures used and to produce the results for the two cases described above. The remaining 
files show the procedures used. 

NB due to the way that the code has been inserted here there may be additional line 
breaks below that should not appear in the code files. 

Case1.pro 
; SBSTA Study of Brazilian Proposal
; Case I: Attribution period is 1890 to 2100
;
; ensure we are running (have compiled) latest versions
.run idl\getemissions
.run idl\ccycle
.run idl\atmchem
.run idl\radforcing
.run idl\plotcpts
;
GetEmissions, 1750, 2100, 1, T, ECO2, ECH4, EN2O, starty=1890, endy=2100
Teco2 = total(Eco2, 1, /cumulative)
Tech4 = total(Ech4, 1, /cumulative)
Ten2o = total(en2o, 1, /cumulative)
WriteCpts, 'NZCase1Ems.txt', 'Cumulative emissions', 'CO2', 'CH4', 'N2O', T, Teco2, Tech4,
Ten2o
;
CCycle, T, ECO2, ACo2, kg=0.07
; GetCO2Obs, Obst, ObsCO2
; Plot, t, aco2, /Ynoz
; Oplot, Obst, Obsco2, Psym=1
;
AtmChem, T, ECH4, EN2O, ACH4, AN2O
; GetCH4Obs, Obst, ObsCH4
; Plot, T, ACH4[*,4], /Ynoz
; Oplot, Obst, ObsCH4, Psym=1
;
WriteCpts, 'NZCase1Cos.txt', 'Concentration changes', 'CO2', 'CH4', 'N2O', T, Aco2, Ach4, An2o
;
RadForcing, ACo2, ACH4, AN2O, RFGHG
GetAerosolRF, T, RFAer, starty=1890, endy=2100
RFTot = RFGHG + RFAer
IRCM, T, RFTot, DTemp
;
; or
; JOOSCM, 3.0, T, RF, DTemp, deltc, delts
;
WriteCpts, 'NZCase1RFT.txt', 'Radiative forcing and temperature changes', 'RFGHG', 'RFAer',
'dTemp', T, RFGHG, RFAer, DTemp
;
plotcpts,T,Dtemp

Case2.pro 
; SBSTA Study of Brazilian Proposal
; Case II: Attribution period is 1890 to 2000
;
; ensure we are running (have compiled) latest versions
.run idl\getemissions
.run idl\ccycle
.run idl\atmchem
.run idl\radforcing
.run idl\plotcpts
;
GetEmissions, 1750, 2100, 1, T, ECO2, ECH4, EN2O, starty=1890, endy=2000



 

 

Teco2 = total(Eco2, 1, /cumulative)
Tech4 = total(Ech4, 1, /cumulative)
Ten2o = total(en2o, 1, /cumulative)
WriteCpts, 'NZCase2Ems.txt', 'Cumulative emissions', 'CO2', 'CH4', 'N2O', T, Teco2, Tech4,
Ten2o
;
CCycle, T, ECO2, ACo2, kg=0.07
; GetCO2Obs, Obst, ObsCO2
; Plot, t, aco2, /Ynoz
; Oplot, Obst, Obsco2, Psym=1
;
AtmChem, T, ECH4, EN2O, ACH4, AN2O
; GetCH4Obs, Obst, ObsCH4
; Plot, T, ACH4[*,4], /Ynoz
; Oplot, Obst, ObsCH4, Psym=1
;
WriteCpts, 'NZCase2Cos.txt', 'Concentration changes', 'CO2', 'CH4', 'N2O', T, Aco2, Ach4, An2o
;
RadForcing, ACo2, ACH4, AN2O, RFGHG
GetAerosolRF, T, RFAer, starty=1890, endy=2000
RFTot = RFGHG + RFAer
IRCM, T, RFTot, DTemp
;
; or
; JOOSCM, 3.0, T, RF, DTemp, deltc, delts
;
WriteCpts, 'NZCase2RFT.txt', 'Radiative forcing and temperature changes', 'RFGHG', 'RFAer',
'dTemp', T, RFGHG, RFAer, DTemp
;

plotcpts,T,Dtemp

Getemissions.pro 
PRO GetEmissions, Yr1, Yr2, Dyr, T, ECO2, ECH4, EN2O, starty=sy, endy=ey
; SBSTA Study for Brazilian Proposal. PHASE II
; TO read standard format files for emission data and do piecewise linear
; ..interpolation onto equi-spaced series of emission values from Yr1 to
; ..Yr2 in steps of Dyr
; Parameters starty, endy specify start and end of attribution period with
; ..all emissions outside this period place in an "unattributed" column.
; Returns arrays ECO2, ECH4, EN2O with 6 columns being emissions for:
; Unattributed, OECD90, REF, ASIA, ALM, Total
; in units of PgC, TgCH4, TgN / yr
;
; 1. Read Table dumped from spreadsheet summarising emissions from
; EDGAR-HYDE 1.4 and SRES A2
;
OPENR, lun, 'EmissionTable.prn', /GET_LUN
aline = ' '
for i = 1, 2 do readf, lun, aline ; Skip headers
e = dblarr(15)

;
; Prefix an emission rate of zero at 1750
te = [1750.0D0]
et = [e]
while not eof(lun) do begin
readf, lun, year, e, format='(16F8.3)'
te = [te, DOUBLE(year)]
et = [et, e]

end
close, lun
free_lun, lun

;
n = n_elements(te)
et = reform(et, 15, n)
et = transpose(et)
print, n-1, 'Emission records read. First and last..'
print, te[1], et[1,*]
print, te[n-1], et[n-1, *]

;
; 2. Do piecewise linear interpolation to equi-spaced time series for all emissions
n = FIX((Yr2-Yr1)/Dyr)+1
t = Dyr*dindgen(n) + DOUBLE(Yr1)
e = dblarr(n, 15)
for i = 0, 14 do begin
e[*,i] = interpol(et[*,i], te, t)



 

 

end
;
; split into species
Fco2 = e[*, 0:4]
Fch4 = e[*, 5:9]
Fn2o = e[*, 10:14]

;
; See below for explanation of this step
BalanceCols, FCO2
BalanceCols, FCH4
BalanceCols, FN2O

;
; Now split out unattributed part of emissions
Eco2 = dblarr(n,6)
Ech4 = dblarr(n,6)
En2o = dblarr(n,6)
If not keyword_set(Sy) then Sy = Yr1
If not keyword_set(Ey) then Ey = Yr2
ix = where( (T LT Sy) OR (T GT Ey), count)
If (count GT 0) then begin
u = total( Fco2[ix,0:3], 2)
Fco2[ix, 0:3] = 0
Eco2[ix,0] = u

;
u = total( Fch4[ix,0:3], 2)
Fch4[ix, 0:3] = 0
Ech4[ix,0] = u

;
u = total( Fn2o[ix,0:3], 2)
Fn2o[ix, 0:3] = 0
En2o[ix,0] = u

end
Eco2[*,1:5] = Fco2
Ech4[*,1:5] = Fch4
En2o[*,1:5] = Fn2o

END
;
==============================================================================================
====
PRO BalanceCols, E
; Past CO2 emissions and future Sulfur emissions do not sum properly as totals include
; bunker fuels which are not in any of the regional columns, also not going to risk
; inputs that do not balance so - use this to rescale the 4 regional columns in each case
; to match the total
sz = size(E)
n = sz[1]
For i = 0, n-1 do begin
zz = total(E[i,0:3])
if (zz ne 0) then E[i,0:3] = E[i,0:3]*(E[i,4]/zz)

end
END
;
==============================================================================================
====
PRO GetAerosolRF, T, RFAer, starty=sy, endy=ey
; SBSTA Study for Brazilian Proposal.
; TO read Aerosol Radiative forcing data.
; Parameters starty, endy specify start and end of attribution period with
; ..all emissions outside this period place in an "unattributed" column.
; Returns array RFAer with 6 columns being emissions for:
; Unattributed, OECD90, REF, ASIA, ALM, Total
; in units of W/m^2
;
; Currently reads data from EDGAR-HYDE and SRES A2 ASF as described in Notes.doc.
;
; Must call GetEmissions first to set up T array with required equi-spaced times.
;
; 1. Read data file
Fpath = 'C:\~MRM\ORG\UNOrgs\FCCC\BrazilProposal\UNFCCC_Assessment\NZResults\Phase2\'
OPENR, lun, Fpath+'AeroRFTable.prn', /GET_LUN
aline = ' '
readf, lun, aline
readf, lun, aline ; skip two lines
readf, lun, DirectFac, IndirectFac
print, DirectFac, IndirectFac
readf, lun, aline
naer = 0



 

 

emis = dblarr(5)
while not eof(lun) do begin
readf, lun, year, emis
year = DOUBLE(year)
if (naer eq 0) then begin
taer = [year]
vaer = [(DirectFac+IndirectFac)*emis]

end else begin
taer = [taer, year]
vaer = [vaer, (DirectFac+IndirectFac)*emis]

end
naer = naer+1

end
close, lun
free_lun, lun
n = n_elements(T)

;
; 2. Pad ends of data - see GetEmissions
; tail off to zero prior to specified range
if (taer[0] gt T[0]) then begin
taer = [T[0], taer]
vaer = [ 0.0D0, 0.0D0, 0.0D0, 0.0D0, 0.0D0, vaer]
naer = naer + 1

end
if (taer[naer-1] lt T[n-1]) then begin
taer = [taer, T[n-1]]
vaer = [vaer, vaer[naer-1,*]]
naer = naer + 1

end
vaer = reform(vaer, 5, naer)
vaer = transpose(vaer)

;
; 3. Do piecewise linear interpolation to equi-spaced time series for all values
RF = dblarr(n,5)
For i = 0, 4 do begin
RF[*,i] = interpol(vaer[*,i], taer, T)

end
;
; Make sure total matches regional sum
Balancecols, RF

;
; Now split out unattributed part of emissions
RFAer = dblarr(n,6)
If not keyword_set(Sy) then Sy = T[0]
If not keyword_set(Ey) then Ey = T[n-1]
ix = where( (T LT Sy) OR (T GT Ey), count)
If (count GT 0) then begin
u = total( RF[ix,0:3], 2)
RF[ix, 0:3] = 0
RFAer[ix,0] = u

end
RFAer[*,1:5] = RF

END

Ccycle.pro 
PRO CCycle, T, ECO2, ACo2, kg=kg, beta=beta, ppm0=ppm0, npp0=npp0
; SBSTA Study for Brazilian Proposal.
; TO calculate CO2 concentration changes from simple carbon cycle model for given CO2
emissions
; Input is T, ECO2, and optionally, kg, beta, ppm0, npp0
; Output is ACO2
; Based on Joos et al Ocean Mixed layer pulse response model.
;
; This version for Phase 2 with attribution. ECO2 and ACO2 are arrays of n x no_regions, where
the
; columns represent amounts attributed to component regions and the total is in ghe last
; column and calculated independently as a check.
;
Forward_Function DpCO2_for_DCs

;
; 1. Get pulse response functions to use.
Get_Response, T, RS, Rdecay

;
; 2. Set parameters for run (Fortunat Joos, Tellus paper and private communication)
;
; ..PgC per ppm of CO2 in atmosphere



 

 

PgC_ppm = 2.123D0
;
; ..Initial (equilibrium) atmospheric pCO2 [ppm]

if not keyword_set(ppm0) then ppm0 = 278.305D0
;
; ..Effective air - sea gas exchange coefficient [Yr^-1]

if not keyword_set(kg) then Kg = 1.0D0/9.6D0
; *** NB. Having to use Kg= 0.07 to reproduce obs ACO2 if nothing else changed. ***
;
; ..Biosphere NPP at start of run (pre-industrial) in PgC/yr

if not keyword_set(npp0) then Npp0 = 60.D0
;
; ..CO2 Fertilization factor

if not keyword_set(beta) then Beta = 0.287D0
; *** NB. 0.287 balances LUC emission of 1.1 PgC/yr in 1980s (Joos et al, 1996)
; 0.38 balances LUC emission of 1.6 PgC/yr in 1980s (IPCC 1994)
;
; 3. Start iterations
sz = size(Eco2)
n = sz[1]
nregions = sz[2]
totcol = nregions-1 ; NB assuming last column has the total emissions
Dyr = T[1] - T[0] ; time step
E = ECO2/PgC_ppm ; emissions as [ppm/yr]
DpCO2a = dblarr(n,nregions) ; to hold atm pCO2 [ppm]
DpCO2s = dblarr(n,nregions) ; to hold sea water pCO2 [ppm]
Fas = dblarr(n,nregions) ; to hold atm-sea flux [ppm/yr]
DCs = dblarr(n,nregions) ; to hold sfc ocean DIC [umol/kg]
Xab = dblarr(nregions)
Fab = dblarr(n,nregions) ; to hold atm-bio flux [ppm/yr]
cume = 0 & cuma = 0 & cumo = 0 & cumb = 0 ; init mass balance integrals
OPENW, lun, 'CCYC.LOG', /get_lun ; Open log file
printf, lun, 'Joos et al Ocean Mixed Layer Pulse Response Model'
printf, lun, 'ppm0= ', ppm0
printf, lun, 'Kg= ', kg
printf, lun, 'NPP0= ', npp0
printf, lun, 'Beta= ', beta
printf, lun, ' T ECO2 x6 DCs x6 DpCO2s x6 Fas x6 Fab x6 DpCO2a x6'

;
; ..MAIN LOOP STARTS HERE
; NB Variables beginning with D hold state values at begin of time step
; Variables beginning with E, F or X are fluxes applying over the time step
For i = 0, n-2 do begin

;
; At i = 0 there has been no change in surface pCO2, else
; ..calculate surface pCO2 change and partition according to air-sea flux in prev step

if (i gt 0) then begin
DpCO2s[i,totcol] = DpCO2_for_DCs(DCs[i,totcol])
delta = DpCO2s[i,totcol] - DpCO2s[i-1,totcol]
drivers = DCs[i,0:totcol-1] - DCs[i-1,0:totcol-1]
if (total(drivers) ne 0.0D0) then begin
DpCO2s[i,0:totcol-1] = DpCO2s[i-1,0:totcol-1] + delta*(drivers/total(drivers) )

end else begin
DpCO2s[i,0:totcol-1] = DpCO2s[i-1,0:totcol-1] ; ?? hopefully does not happen ??

end
end

;
; ..Calculate exchange fluxes for time step i

Fas[i,*] = Kg*(DpCO2a[i,*] - DpCO2s[i,*]) ; NB in [ppm / yr]
;
; ..for biosphere partition according to increase in atm increment in prev step
; NB start by storing the increment in NPP driven flux for housekeeping reasons

delta = Npp0*Beta*ALOG(1.0D0+(DpCO2a[i,totcol]/ppm0))/PgC_ppm - Xab[totcol]
Xab[totcol] = Xab[totcol] + delta ; [ppm / yr]
if (i gt 0) then begin
drivers = DpCO2a[i,0:totcol-1] - DpCO2a[i-1,0:totcol-1]
if (total(drivers) ne 0.0D0) then begin
Xab[0:totcol-1] = Xab[0:totcol-1] + delta*(drivers/total(drivers) )

end
end
for k=0,totcol do Fab[i,k] = Fab[i,k] + Xab[k]

;
; ..Accumulate committments of these fluxes to all future times for DCs and Fab

IX = INDGEN(n-i-1)+1 ; 1..(n-i-1), so IX+i = (i+1)..(n-
1)

for k=0,totcol do DCs[IX+i,k] = DCs[IX+i,k] + Fas[i,k]*RS[IX]
for k=0,totcol do Fab[IX+i,k] = Fab[IX+i,k] - Xab[k]*Rdecay[IX]



 

 

;
; ..Add changes to atmosphere

DpCO2a[i+1,*] = DpCO2a[i,*] + Dyr*(E[i,*] - Fas[i,*] - Fab[i,*])
;
; ..cumulative mass balance diagnostics for
; emission, change in atm, change in ocean, change in biosphere
; [all converted to ppm equivalent]

cume = cume + E[i,totcol]*Dyr
cuma = DpCO2a[i+1,totcol]
cumo = cumo + Fas[i,totcol]*Dyr
cumb = cumb + Fab[i,totcol]*Dyr
printf, lun, T[i+1], E[i,*], DCs[i,*], DpCO2s[i,*], Fas[i,*], Fab[i,*], DpCO2a[i+1,*],

Format='(1x,37F10.3)'
end

; ..END MAIN LOOP
;
close, lun
free_lun, lun

;
; The take home result is....
ACO2 = DpCO2a

;
; some helpful diagnostics
ix80s = where((t ge 1980) and (t lt 1990))
ix90s = where((t ge 1990) and (t lt 2000))
print, 'Mean emissions in 1980s, 1990s', PgC_ppm*total(E(ix80s,totcol))/n_elements(ix80s), $

PgC_ppm*total(E(ix90s,totcol))/n_elements(ix90s)
print, 'Mean ocean uptake in 1980s, 1990s',

PgC_ppm*total(Fas(ix80s,totcol))/n_elements(ix80s), $
PgC_ppm*total(Fas(ix90s,totcol))/n_elements(ix90s)

print, 'Mean biosphere uptake in 1980s, 1990s',
PgC_ppm*total(Fab(ix80s,totcol))/n_elements(ix80s), $

PgC_ppm*total(Fab(ix90s,totcol))/n_elements(ix90s)
print, 'CO2 concentrations, 1980, 1990, 2000', Aco2(ix80s[0],totcol), Aco2(ix90s[0],totcol),

$
Aco2[where(t eq 2000),totcol]

;
; check that attributed changes add to total change
temp = DpCO2a[*,totcol] - total( DpCO2a[*,0:totcol-1], 2 )
print, 'Min and max closure error on DpCO2a ', min(temp), max(temp, imax), imax
temp = DpCO2s[*,totcol] - total( DpCO2s[*,0:totcol-1], 2 )
print, 'Min and max closure error on DpCO2s ', min(temp), max(temp, imax), imax
temp = Fas[*,totcol] - total( Fas[*,0:totcol-1], 2)
print, 'Min and max closure error on Fas ', min(temp), max(temp, imax), imax
temp = Fab[*,totcol] - total( Fab[*,0:totcol-1], 2)
print, 'Min and max closure error on Fab ', min(temp), max(temp, imax), imax

END
;=============================================================================================
==
PRO Get_Response, T, RS, Rdecay
; Set up response functions
; From Joos et al (1996) Tellus paper
DT = T - T[0] ; basis time for response functions same length as T

starting at t=0
Dyr = DT[1]
n = n_elements(T)

;
; OCEAN --------------------------------------------------------------------------------------
-
; Ocean mixed layer response function (HILDA version)
RS = dblarr(n)
ix = Where(DT lt 2.0D0)
T1 = DT[ix]
RS[ix] = 0.12935D0 + 0.21898D0*exp(-T1/0.034569D0) + 0.17003D0*exp(-T1/0.26936D0) + $

0.24071D0*exp(-T1/0.96083D0) + 0.24093D0*exp(-T1/4.9792D0)
ix = Where(DT ge 2.0D0)
T1 = DT[ix]
RS[ix] = 0.022936D0 + 0.24278D0*exp(-T1/1.2679D0) + 0.13963D0*exp(-T1/5.2528D0) + $

0.089318D0*exp(-T1/18.601D0) + 0.037820D0*exp(-T1/68.736D0) + 0.035549D0*exp(-
T1/232.30D0)
;
; Scale to umol/kg using...
PgC_ppm = 2.123D0 ; = PgC per ppm of CO2 in atmosphere
gCperMole = 12.0113D0
Hs = 75.0D0 ; Ocean mixed layer depth [m]
Aoc = 3.62D14 ; Ocean area [m^2]
dens = 1.0265D3 ; sea water density [kg/m^3]



 

 

; calculate change in ocean sfc Carbon [umol/kg] for 1 ppm removal from atm
umolperkg_ppm = (1D21*PgC_ppm/gCpermole)/(dens*Hs*Aoc)
RS = RS*umolperkg_ppm*Dyr

;
; BIOSPHERE ----------------------------------------------------------------------------------
-
; Biosphere response function
Rdecay = dblarr(n)
Rdecay = 0.7021D0*exp(-0.35D0*DT) + 0.01341D0*exp(-DT/20.D0) - 0.7185D0*exp(-0.4583D0*DT) +

$
0.002932D0*exp(-0.01D0*DT)

Rdecay = Rdecay*Dyr ; NB this corrects Rdecay to have unit area
;
;** Quick checks
;** ix = where((DT lt 2.01) and (DT gt 1.99))
;** ix1 = where((DT lt 5.01) and (DT gt 4.99))
;** print, Dyr, RS[ix], Rdecay[ix1]
;** plot, DT, RS
;** oplot, DT, Rdecay
END
;=============================================================================================
==
FUNCTION DpCO2_for_DCs, DCs
; Returns change in pCO2A from equilibrium corresponding to change in
; Ocean mixed layer carbon from equilibrium. See Joos et al, Tellus paper 1996.
; DCs in [umol/kg], and return value in [ppm].

TC = 18.1716D0 ; Effective Ocean temperature for carbonate chemistry [C]
a1 = (1.5568D0 - 1.3993D-2*TC)
a2 = (7.4706D0 - 0.20207D0*TC)*1.D-3
a3 = -(1.2748D0 - 0.12015D0*TC)*1.D-5
a4 = (2.4491D0 - 0.12639D0*TC)*1.D-7
a5 = -(1.5468D0 - 0.15326D0*TC)*1.D-10
RETURN, DOUBLE(DCs*(a1+DCs*(a2+DCs*(a3+DCs*(a4+DCs*a5)))))

END
;=============================================================================================
==
PRO GetCO2Obs, Obst, ObsCO2
; SBSTA Study for Brazilian Proposal.
; TO read Composite of Law Dome CSIRO (David Etheridge) and SIO MLO (C.D. Keeling)
; CO2 data ; for comparison with integrated values from Cycle procedure.
;
; 1. Read my composite file and retrieve year and CO2 concentration
OPENR, lun,

'C:\~MRM\ORG\UNOrgs\FCCC\BrazilProposal\UNFCCC_Assessment\Obs\CombinedCO2Data.csv', /GET_LUN
aline = ' '
for i = 1, 3 do readf, lun, aline
n = 0
while not eof(lun) do begin
readf, lun, year, val
print, year, val
if (n eq 0) then begin
Obst = [year]
ObsCO2 = [val]

end else begin
Obst = [Obst, year]
ObsCO2 = [ObsCO2, val]

end
n = n+1

end
close, lun
free_lun, lun

END

 

Atmchem.pro 
PRO AtmChem, T, ECH4, EN2O, ACH4, AN2O
; SBSTA Study for Brazilian Proposal.
; TO do atmospheric chemistry calculations.
; ..Currently CHANGES in
; CH4 concentrations from emissions for a fixed lifetime, and
; .. N2O concentrations from emissions for a fixed lifetime
;



 

 

; This version for Phase 2 with attribution. ECO2 and ACO2 are arrays of n x no_regions, where
the
; columns represent amounts attributed to component regions and the total is in ghe last
; column and calculated independently as a check.
;
sz = size(Ech4)
n = sz[1]
nregions = sz[2]
totcol = nregions-1 ; NB assuming last column has the total emissions
Dyr = T[1] - T[0]

;
; 1. Integrate ODE for CH4 with fixed lifetime
; Constants for CH4 determined offline to give best fit to Obs with
; EDGAR-HYDE emissions (really need time varying lifetime though!)
; BaseCH4 = 595.4D0 ; ppb - see Notes
TauCH4 = 10.0D0 ; adjusted to get 1745 ppb in 1998
LamCH4 = 1.0D0/TauCH4 ; inverse lifetime in years^-1
ScaleCH4 = 2.78D0 ; TgCH4 per ppb (this is TAR Chap 4 value)

; ECH4Nat = 146.95D0
Ach4 = dblarr(n,nregions)

; Ach4[0,*] = BaseCH4
; exact integration over each time step in following loop uses two constants:
; decay over 1 time step, and
; factor multiplying average emission rate over the time step, as follows
Decay = exp(-LamCH4*Dyr)
Accum = (1.0D0 - Decay)/(LamCH4*ScaleCH4)
for i = 1, n-1 do begin

; ach4[i,*] = ach4[i-1,*]*Decay + (ECH4Nat + Ech4[i-1,*])*Accum
ach4[i,*] = ach4[i-1,*]*Decay + Ech4[i-1,*]*Accum

end
;
; 2. Now repeat for N2O
; Constants merely taken from TAR etc - NOT optimized to observations.
; BaseN2O = 266D0
TauN2O = 114D0
LamN2O = 1.0D0/TauN2O
ScaleN2O = 4.8

; EN2ONat = 11.2
An2o = dblarr(n,nregions)

; An2o[0,*] = BaseN2O
Decay = exp(-LamN2O*Dyr)
Accum = (1.0D0 - Decay)/(LamN2O*ScaleN2O)
for i = 1, n-1 do begin

; an2o[i,*] = an2o[i-1,*]*Decay + (EN2ONat + En2o[i-1,*])*Accum
an2o[i,*] = an2o[i-1,*]*Decay + En2o[i-1,*]*Accum

end
;
END
;=============================================================================================
==
PRO GetCH4Obs, Obst, ObsCH4
; SBSTA Study for Brazilian Proposal.
; TO read Composite of CSIRO (David Etheridge) and GLOBALVIEW CH4 data
; for comparison with integrated values from emissions via "AtmChem" procedure.
;
; 1. Read my composite file and retrieve year and CH4 concentration
OPENR, lun, 'C:\~MRM\DAT\AtmChem\CH4\CompositeAtmCH4.csv', /GET_LUN
aline = ' '
for i = 1, 5 do readf, lun, aline
n = 0
while not eof(lun) do begin
readf, lun, year, val

; print, year, val
if (n eq 0) then begin
Obst = [year]
ObsCH4 = [val]

end else begin
Obst = [Obst, year]
ObsCH4 = [ObsCh4, val]

end
n = n+1

end
close, lun
free_lun, lun

END



 

 

Radforcing.pro 
PRO RadForcing, ACo2, ACH4, AN2O, RFGHG
; SBSTA Study for Brazilian Proposal.
; TO calculate radiative forcing due to CO2 and CH4 changes
; Formula are from the TAR
;
BaseCO2 = 278.D0 ; pre-industrial CO2 level
BaseCH4 = 700.D0 ; pre-industrial CH4 level
BaseN2O = 270.D0 ; pre-industrial N2O level
sz = size(ACO2)
n = sz[1]
nregions = sz[2]
totcol = nregions-1
RFCO2 = dblarr(n,nregions)

;
; ..Total CO2 Forcing
RFCO2[*,totcol] = 5.35*ALOG(1 + Aco2[*,totcol]/BaseCO2 )

;
; ..Partition according to changes in CO2 by region
For i = 1, n-1 do begin
delta = RFCO2[i,totcol] - RFCO2[i-1,totcol]
drivers = ACO2[i,0:totcol-1] - ACO2[i-1,0:totcol-1]
if (total(drivers) gt 0) then $

RFCO2[i,0:totcol-1] = RFCO2[i-1,0:totcol-1] + delta*(drivers/total(drivers))
end

;
; ..Total CH4 Forcing
Fnow = 0.47*ALOG(1 + 2.01D-5*( (BaseCH4+ACH4[*,totcol])*BaseN2O)^0.75 + $

5.31D-15*(BaseCH4+ACH4[*,totcol])*((BaseCH4+ACH4[*,totcol])*BaseN2O)^1.52)
Fthen = 0.47*ALOG(1 + 2.01D-5*(BaseCH4*BaseN2O)^0.75 + $

5.31D-15*BaseCH4*(BaseCH4*BaseN2O)^1.52)
RFCH4 = dblarr(n,nregions)
RFCH4[*,totcol] = 0.036*(SQRT(BaseCH4+ACH4[*,totcol]) - SQRT(BaseCH4)) - (Fnow - Fthen)

;
; ..Partition according to changes in CH4 by region
For i = 1, n-1 do begin
delta = RFCH4[i,totcol] - RFCH4[i-1,totcol]
drivers = ACH4[i,0:totcol-1] - ACH4[i-1,0:totcol-1]
if (total(drivers) gt 0) then begin
RFCH4[i,0:totcol-1] = RFCH4[i-1,0:totcol-1] + delta*(drivers/total(drivers))

end
end

;
; ..Total N2O Forcing
Fnow = 0.47*ALOG(1 + 2.01D-5*(BaseCH4*(BaseN2O+AN2O[*,totcol]))^0.75 + $

5.31D-15*BaseCH4*(BaseCH4*(BaseN2O+AN2O[*,totcol]))^1.52)
RFN2O = dblarr(n,nregions)
RFN2O[*,totcol] = 0.12*(SQRT(BaseN2O+AN2O[*,totcol]) - SQRT(BaseN2O)) - (Fnow - Fthen)

;
; ..Partition according to changes in N2O by region
For i = 1, n-1 do begin
delta = RFN2O[i,totcol] - RFN2O[i-1,totcol]
drivers = AN2O[i,0:totcol-1] - AN2O[i-1,0:totcol-1]
if (total(drivers) gt 0) then begin
RFN2O[i,0:totcol-1] = RFN2O[i-1,0:totcol-1] + delta*(drivers/total(drivers))

end
end

;
; ..Return sum of forcings
RFGHG = RFCO2 + RFCH4 + RFN2O

END
;=============================================================================================
==
PRO IRCM, T, RF, DTemp
; SBSTA Study for Brazilian Proposal.
; TO calculate Temperature change according to an impulse response climate model
; defined by a convolution integral of a two-exponential kernel and radiative forcing
; determined so as to match results of the HadCM3 model. See Jason Lowe's contribution to the
; web pages for this study.
;
; 1. Define the response function (multiply by time step scales to unit area)
Trange = T - T[0]
Rtemp = (0.59557D0/8.4007D0)*exp(-Trange/8.4007D0) + (0.40443D0/409.54D0)*exp(-

Trange/409.54D0)
Rtemp = Rtemp*(T[1]-T[0])

;



 

 

; 2. convolution integral
sz = size(RF)
n = sz[1]
nregions = sz[2]
DTemp = dblarr(n,nregions)
for i = 0, n-1 do begin
IX = INDGEN(n-i)
for k = 0, nregions-1 do DTemp[IX+i,k] = DTemp[IX+i,k] + RF[i,k]*Rtemp[IX]

end
;
; 2. Scaling set by final RF value and climate sensitivity
; Sens = 0.809D0 ; original climate sensitivity in degC/(w/m^2)
; tweaked value to improve fit to recent temperatures
Sens = 1.1D0
Dtemp = Sens*Dtemp

END
;=============================================================================================
==
PRO JoosCM, T2x, T, RF, DTemp, deltc, delts, continent=continent
; SBSTA Study for Brazilian Proposal.
; TO calculate Temperature change according to Fortunat Joos's energy balance model
; using the ocean mixed layer response function applied to heat.
; T2x is equilibrium warming for 2 x CO2-equivalent
; if "/continent" specified then implements version with separate continental and sea
; ..temperatures in which case if "dtc" and or "dts" parameters set then continental
; ..and sea temperature arrays returned there.
;
; NB This is essentially a linear model connecting RF to DTemp - so do attribution to
; regions by treating their radiative forcing independently.
;
; 1. Define constants
cw = 4000.D0 ; specific heat of seawater [J/(kg.K)]
Hs = 75.0D0 ; Ocean mixed layer depth [m]
dens = 1.0265D3 ; sea water density [kg/m^3]
foc = 0.71D0 ; fraction of surface covered by sea
secperyear=3600.D0*24.D0*365.25D0 ; seconds per year

;
if keyword_set(continent) then begin

; .. following from Fortunat's code (not published) apparently corresponds to atmospheric
; .. relaxation time of 8 days
;***** THIS OPTION NOT WORKING 25 AUG 2002 *******

kcs = 7.2D0*secperyear ; exchange coefficient continent to sea
[J/yr.m^2.K]

alpha = 1.3D0 ; ratio of continent to sea equil
temperature
end else begin
kcs = 0.D0
alpha = 1.D0

end
ksc = kcs*(1.D0 - foc)/foc ; this for energy balance bt continent and sea
T2xS = T2x/(alpha*(1.D0 - foc) + foc)
T2xC = alpha*T2xS

;
; NB working with time units = year so scale radiative forcing to J/yr.m^2
RFy = RF*secperyear
RF2x = 5.35*ALOG(2)*secperyear ; radiative forcing for 2 x CO2
Dyr = T[1] - T[0] ; time step for normalizing response fn
Get_response, T, RS, Rdecay ; get response functions - only use RS for ocean mixed

layer
;
; set up arrays
sz = size(RF)
n = sz[1]
nregions = sz[2]
delts = dblarr(n,nregions)
deltc = dblarr(n,nregions)
Fas = dblarr(nregions)

;
; 2. Loop over all time steps...
for i = 1, (n-1) do begin

; ..calculate heat flux into mixed layer this time step
Fas = RFy[i,*] - (RF2x/T2xS)*delts[i-1,*] + ksc*(deltc[i-1,*] - alpha*delts[i-1,*])
Fas = Fas*Dyr/(cw*Hs*dens) ; scaling to temperature units after convolution

integral
; ..accumulate forward committment of this flux to all future times

IX = INDGEN(n-i)
for k = 0, nregions-1 do delts[IX+i,k] = delts[IX+i,k] + Fas[k]*RS[IX]



 

 

; ..deltc array calculated from delts one element at a time
deltc[i,*] = (RFy[i,*] + kcs*delts[i,*])/(RF2x/T2xC + kcs/alpha)

;
print, 'deltc, delts', deltc[i,*], delts[i,*], format='(a,12f8.3)'

end
;
Dtemp = foc*delts + (1.D0-foc)*deltc

END
PRO GetTempObs, Obst, ObsTemp
; SBSTA Study for Brazilian Proposal.
; TO read UEA data file of monthly temperatures but only returns time
; series of annual average anomalies.
;
; 1. Read file and retrieve year and Temperature
Path = 'C:\~MRM\ORG\UNOrgs\FCCC\BrazilProposal\UNFCCC_Assessment\Obs\'
OPENR, lun, Path+'global_dat.txt', /GET_LUN
aline = ' '
for i = 1, 17 do readf, lun, aline
n = 0
for i = 1, 145 do begin
readf, lun, year, val, format='(f4.0, 84x, f7.2)'

;
print, year, val
if (n eq 0) then begin
Obst = [year]
ObsTemp = [val]

end else begin
Obst = [Obst, year]
ObsTemp = [ObsTemp, val]

end
n = n+1

end
close, lun
free_lun, lun

END

Plotcpts.pro 
PRO PlotCpts,t,y, ymin=ymin, ymax=ymax, title=title
; to plot a time series graph for the components in y
; as a stacked area plot.
;
device, decomposed=0 ; assuming we are set to WIN device
loadct, 27 ; Pick EOS B colors

;
sz = size(y)
n = sz[1]
nregions = sz[2]
totcol = nregions-1
if not keyword_set(ymin) then ymin = min(y)
if not keyword_set(ymax) then ymax = max(y)

;
plot,t,y[*,totcol],/ynoz, $
xrange=[min(t), max(t)], xsty=1, $
yrange=[ymin, ymax], ysty=1 ; this to set axes and scales

base = dblarr(n) ; base for next area graph
for i = 0, totcol-1 do begin
polyfill, [t, reverse(t)], [base+y[*,i], reverse(base)], $
color=fix((i+0.5)*255/nregions), noclip=0

base = base+y[*,i]
end

;
; then replot boundary lines to clarify where negative areas occur
base = dblarr(n)
for i = 0, totcol-1 do begin
oplot, t, base+y[*,i], color=fix((i+0.5)*255/nregions)
base = base + y[*,i]

end
; legend
lx0 = 0.18
lx1 = 0.23
ly0 = 0.4
dy = 0.05
name = ['Unattributed', 'OECD90', 'REF', 'ASIA', 'ALM', 'Total']
for i = 0, totcol-1 do begin
c = fix((i+0.5)*255/nregions)
ly1 = ly0 + (i+1)*dy



 

 

polyfill, [lx0, lx1, lx1, lx0], [ly1-dy, ly1-dy, ly1, ly1], $
/normal, color=c

xyouts, lx1+0.2*dy, ly1-0.7*dy, name[i], /normal, color=c
end

;
; if title parameter set write that string
if keyword_set(title) then $
xyouts, 0.5, 0.7, title, align=0.5, charsize=1.4, /normal

;
END
PRO PlotLines,t,y, ymin=ymin, ymax=ymax
; to plot a time series graph for the components in y
; as a series of lines
;
device, decomposed=0 ; assuming we are set to WIN device
loadct,27 ; picking EOS B color table

;
sz = size(y)
n = sz[1]
nregions = sz[2]
totcol = nregions-1
if not keyword_set(ymin) then ymin = min(y[*,0:totcol-1])
if not keyword_set(ymax) then ymax = max(y[*,0:totcol-1])

;
plot,t,y[*,0],/ynoz, /nodata, $
xrange=[min(t), max(t)], xsty=1, $
yrange=[ymin, ymax], ysty=1

for i = 0, totcol-1 do begin
oplot, t, y[*,i], color=fix((i+0.5)*255/nregions), thick=2

end
;
; legend
lx0 = 0.18
lx1 = 0.23
ly0 = 0.4
dy = 0.05
name = ['Unattributed', 'OECD90', 'REF', 'ASIA', 'ALM', 'Total']
for i = 0, totcol-1 do begin
c = fix((i+0.5)*255/nregions)
ly1 = ly0 + (i+1)*dy
polyfill, [lx0, lx1, lx1, lx0], [ly1-dy, ly1-dy, ly1, ly1], $
/normal, color=c

xyouts, lx1+0.2*dy, ly1-0.7*dy, name[i], /normal, color=c
end

END
PRO WriteCpts, fname, title, par1, par2, par3, t, v1, v2, v3
; writes a text file with values for 3 variables, v1, v2, and v3.
; fname is the file name to be created; title is a title line;
; par1, par2, and par3 are the variable names (headings for the
; 6 regional components are supplied automatically).
;
openw, lun, fname, /get_lun
printf, lun, Title
printf, lun, 'Year', par1, par1, par1, par1, par1, par1, $
par2, par2, par2, par2, par2, par2, $
par3, par3, par3, par3, par3, par3, format='(19a11)'

printf, lun, ' ', 'Unattr', 'OECD90', 'REF', 'ASIA', 'ALM', 'Total', $
'Unattr', 'OECD90', 'REF', 'ASIA', 'ALM', 'Total', $
'Unattr', 'OECD90', 'REF', 'ASIA', 'ALM', 'Total', format='(19a10)'

for i = 0, 350 do begin
printf, lun, t[i], v1[i,*], v2[i,*], v3[i,*], format='(19f11.4)'

end
close, lun
free_lun, lun

END

